

**graph twoway pccapsym** — Paired-coordinate plot with spikes and marker symbols

[Description](#)  
[Options](#)

[Quick start](#)  
[Remarks and examples](#)

[Menu](#)  
[Also see](#)

[Syntax](#)

## Description

A paired-coordinate capped-symbol plot draws a spike (or line) for each observation in the dataset and caps these spikes with a marker symbol at each end. The line starts at the coordinate ( $y1var, x1var$ ) and ends at the coordinate ( $y2var, x2var$ ), and both coordinates are designated with a marker.

## Quick start

Paired-coordinate plot with spikes from ( $y1, x1$ ) to ( $y2, x2$ ) and marker symbols

```
twoway pccapsym y1 x1 y2 x2
```

Horizontal paired-coordinate plot

```
twoway pccapsym y1 x1 y2 x2, horizontal
```

Specify that starting points should be labeled using the values of `labvar`

```
twoway pccapsym y1 x1 y2 x2, mlabel(labvar)
```

As above, but specify labels appear at the end points

```
twoway pccapsym y1 x1 y2 x2, mlabel(labvar) headlabel
```

As above, but specify marker symbols are squares

```
twoway pccapsym y1 x1 y2 x2, mlabel(labvar) msymbol(square)
```

## Menu

Graphics > Twoway graph (scatter, line, etc.)

## Syntax

```
twoway pccapsym y1var x1var y2var x2var [if] [in] [, options]
```

<i>options</i>	Description
<i>line_options</i>	change look of spike lines
<i>marker_options</i>	change look of markers (color, size, etc.)
<i>marker_label_options</i>	add marker labels; change look or position
<b>headlabel</b>	label second coordinate, not first
<b>vertical</b>	orient plot naturally; the default
<b>horizontal</b>	orient plot transposing <i>y</i> and <i>x</i> values
<i>axis_choice_options</i>	associate plot with alternative axis
<i>twoway_options</i>	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

All explicit options are *rightmost*, except **headlabel**, **vertical**, and **horizontal**, which are *unique*; see [G-4] **concept: repeated options**.

## Options

*line\_options* specify the look of the lines used to draw the spikes, including pattern, width, and color; see [G-3] **line\_options**.

*marker\_options* specify how the markers look, including shape, size, color, and outline; see [G-3] **marker\_options**. The same marker is used on both ends of the spikes.

*marker\_label\_options* specify if and how the markers are to be labeled; see [G-3] **marker\_label\_options**.

**headlabel** specifies that labels be drawn on the markers of the (*y2var*,*x2var*) points rather than on the markers of the (*y1var*,*x1var*) points. By default, when the **mlabel()** option is specified, labels are placed on the points for the first two variables—*y1var* and *x1var*. **headlabel** moves the labels from these points to the points for the second two variables—*y2var* and *x2var*.

**vertical** and **horizontal** specify whether the *y* and *x* coordinates are to be swapped before plotting—**vertical** (the default) does not swap the coordinates, whereas **horizontal** does.

These options are rarely used when plotting only paired-coordinate data; they can, however, be used to good effect when combining paired-coordinate plots with range plots, such as **twoway rspike** or **twoway rbar**; see [G-2] **graph twoway rspike** and [G-2] **graph twoway rbar**.

*axis\_choice\_options* associate the plot with a particular *y* or *x* axis on the graph; see [G-3] **axis\_choice\_options**.

*twoway\_options* are a set of common options supported by all **twoway** graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over **by()** groups, and change some advanced settings. See [G-3] **twoway\_options**.

## Remarks and examples

**stata.com**

Remarks are presented under the following headings:

*Basic use 1*  
*Basic use 2*

## Basic use 1


We have longitudinal data from 1968 and 1988 on the earnings and total experience of U.S. women by occupation.

```
. use http://www.stata-press.com/data/r15/nlswide1
(National Longitudinal Survey. Young Women 14-26 years of age in 1968)
. list occ wage68 ttl_exp68 wage88 ttl_exp88
```

	occ	wage68	ttl_e~68	wage88	ttl_e~88
1.	Professionals	6.121874	.860618	10.94776	14.11177
2.	Managers	5.426208	1.354167	11.53928	13.88886
3.	Sales	4.836701	.9896552	7.290306	12.62823
4.	Clerical/unskilled	4.088309	.640812	9.612672	11.08019
5.	Craftsmen	4.721373	1.091346	7.839769	12.64364
6.	Operatives	4.364782	.7959284	5.893025	11.99362
7.	Transport	1.987857	.5247414	3.200494	8.710394
8.	Laborers	3.724821	.775966	5.264415	10.56182
9.	Other	5.58524	.8278245	8.628641	12.78389

We graph a spike with symbols capping the end to show the movement from 1968 values to 1988 values for each observation (each occupation):

```
. twoway pccapsym wage68 ttl_exp68 wage88 ttl_exp88
```




For a better presentation of these data, see [Advanced use](#) in [\[G-2\] graph twoway pccapsym](#); the comments there about combining plots apply equally well to `pccapsym` plots.

## Basic use 2

We can draw both the edges and nodes of network diagrams by using `twoway pccapsym`.

```
. use http://www.stata-press.com/data/r15/network1  
. twoway pccapsym y_c x_c y_l x_l
```



Again, a better presentation of these data can be found in [\[G-2\] graph twoway pcspike](#) under *Advanced use 2*.

## Also see

[\[G-2\] graph twoway](#) — Twoway graphs

[\[G-2\] graph twoway line](#) — Twoway line plots

[\[G-2\] graph twoway rcapsym](#) — Range plot with spikes capped with marker symbols

[\[G-2\] graph twoway pcarrow](#) — Paired-coordinate plot with arrows

[\[G-2\] graph twoway pci](#) — Twoway paired-coordinate plot with immediate arguments

[\[G-2\] graph twoway pscatter](#) — Paired-coordinate plot with markers

[\[G-2\] graph twoway pcspike](#) — Paired-coordinate plot with spikes